
turtles-own 
[ 
  water     ;; Amount of stored water 
  sugar     ;; Amount of stored sugar 
  adjacent  ;; Holds the identity of the adjacent turtles when nutrients are being shared 
] 
 
patches-own 
[ 
  moisture  ;; Amount of water in the soil 
  light     ;; Amount of light available for the leaves to turn into sugar 
] 
 
to setup-patches 
  clear-all 
  ask patches 
  [ 
    ifelse pycor > 0 
    [ ;; Allocate Light 
      ifelse (random (world-width ^ 2)) < (nutrient-density * world-width) 
      [ set light random nutrient-concentration ] 
      [ set light 5 ] 
    ] 
    [ ;; Allocate Moisture 
      ifelse (random (world-width ^ 2)) < (nutrient-density * world-width) 
      [ set moisture random nutrient-concentration ] 
      [ set moisture 5 ] 
    ] 
  ] 
  diffuse-light 
  diffuse-moisture 
  ask patches 
  [ 
    ifelse pycor > 0 
    [ 
      set pcolor scale-color yellow light 14 -1 
      set moisture 5  ;; No moisture in the light area 
    ] 
    [ 
      set pcolor scale-color blue moisture 14 -1 
      set light 5  ;; No light in the moisture area 
    ] 
    ;; draw the ground 



    if pycor = 0 and abs pxcor > 2 
    [ set pcolor gray ] 
  ] 
  reset-ticks 
end 
 
to diffuse-light 
  diffuse light 0.1 
  if max [light] of patches > 15 
  [ diffuse-light ] 
end 
 
to diffuse-moisture 
  diffuse moisture 0.1 
  if max [moisture] of patches > 15 
  [ diffuse-moisture ] 
end 
 
 
to setup-plant 
  set-default-shape turtles "circle" 
  ;; Kill the old Plant 
  ask turtles [ die ] 
  ;; Create the new Plant 
  create-turtles 1 
  [ 
    set color brown 
    set sugar 5000 
    set water 5000 
    set heading 0 
    hatch 1 
    [ 
      set color green fd 1 
    ] 
  ] 
end 
 
to grow 
  ask turtles [ 
  ;; Get Nutrients from environment 
  ifelse color = green 
  [ set sugar sugar + light ] 
  [ set water water + moisture ] 



  ;; Grow Plant 
  if random 100 < 1 
  [ 
    hatch 1 
    [ 
      move 
      ;; Five Conditions under which the new growth should be aborted 
      if sum [count turtles-here] of neighbors >= 3 [ die ]  ;; Overcrowding 
      if any? other turtles-here [ die ]  ;; Overlapping 
      if color = green and pycor < 1 [ die ]  ;; Leaves Underground 
      if color = brown and pycor > 0 [ die ]  ;; Roots Aboveground 
      if pcolor = gray [ die ]  ;; In the ground 
      set sugar 1 
      set water 1 
    ] 
  ] 
  share-with-gs 
  ;; Use Resources 
  set sugar sugar - 0.1 
  set water water - 0.1 
  if sugar <= 0 or water <= 0 [ die ] 
  ] 
  tick 
end 
 
to move 
  ifelse cactus? 
  [ ;; Plant grows up and down only 
    set heading 180 * random 2 
    rt 30 - 30 * random 3 
  ] 
  [ ;; Plant grows in all directions 
    rt random-float 360 
  ] 
  ;; if this is the edge of the world obviously don't grow there. 
  ifelse can-move? 1 
  [ fd 1 ] 
  [ die ] 
end 
 
to share-with-gs 
  set adjacent nobody 
  if any? turtles-at 1 1 



  [ set adjacent one-of turtles-at 1 1 
    share-up 
  ] 
  if any? turtles-at 0 1 
  [ set adjacent one-of turtles-at 0 1 
    share-up 
  ] 
  if any? turtles-at -1 1 
  [ set adjacent one-of turtles-at -1 1 
    share-up 
  ] 
  if any? turtles-at 1 0 
  [ set adjacent one-of turtles-at 1 0 
    share-side 
  ] 
  if any? turtles-at 1 -1 
  [ set adjacent one-of turtles-at 1 -1 
    share-down 
  ] 
  if any? turtles-at 0 -1 
  [ set adjacent one-of turtles-at 0 -1 
    share-down 
  ] 
  if any? turtles-at -1 -1 
  [ set adjacent one-of turtles-at -1 -1 
    share-down 
  ] 
end 
 
to share-up 
  let old-water water 
  set water 0.95 * water + 0.02 * [water] of adjacent 
  ask adjacent [ set water 0.98 * water + 0.02 * old-water ] 
end 
 
;; Nutrients are shared equally, but the sharing is executed by the left turtle 
to share-side 
  let old-water water 
  set water 0.95 * water + 0.05 * [water] of adjacent 
  ask adjacent [ set water 0.95 * water + 0.05 * old-water ] 
  let old-sugar sugar 
  set sugar 0.95 * sugar + 0.05 * [sugar] of adjacent 
  ask adjacent [ set sugar 0.95 * sugar + 0.05 * old-sugar ] 



end 
 
to share-down 
  let old-sugar sugar 
  set sugar 0.95 * sugar + 0.02 * [sugar] of adjacent 
  ask adjacent [ set sugar 0.98 * sugar + 0.05 * old-sugar ] 
end 

